


SAS Examples

Mediation

Background: Mediation modeling is a powerful analytical tool that can be used to explain the nature of the relationship among three or more variables. In addition to showing this simple relationship, it can be used to show how a variable mediates the relationship between levels of intervention and outcome. Recent works provide a statistical test for the mediation path known as the "Sobel Test." While interactive web-based and stand alone methods exist for computing the Sobel Test, it is useful to have a program that will run the required regression analyses and compute the test statistic automatically.



You can download the data files for the SAS example as a Winzip file by clicking on program files.

Syntax:

```
/* ***** */
/* This program estimates the percentage of the total effect that is mediated*/
/* and the ratio of the indirect to the direct effect by using the Sobel test */
/* Written by William Dudley PhD and Jose Benezillo MA */
/* University of Utah College of Nursing */
/* 12/13/2002 */
/* ***** */
libname sobel 'a:\mediation\sas\example Data'; /* Change path */
run;

data test;
    set sobel.example; /* Change data name */
    iv= pain; /* your IV */
    mediator= function; /*your MEDIATOR */
    dv= depress; /* your DV */
run;

/* ***** */
/* Regression analysis: IV predicting DV */
/* IV predicting MEDIATOR */
/* IV and MEDIATOR predicting DV */
/* ***** */

proc reg data=test;
    model dv=iv;
    model mediator=iv;
    model dv=iv mediator;
    ods output ParameterEstimates=reg1; /* Saving estimates into a data set */
run;
quit;

proc format;
    value $model 'MODEL1' = 'Regressing IV on DV'
                'MODEL2' = 'Regressing IV on Mediator'
                'MODEL3' = 'Regressing IV and Mediator on DV';
run;

/* Here we select only the statistics required to compute the Sobel test */

data sobel;
    set reg1;
```

```

    if model = 'MODEL2' and variable = 'iv' then a = Estimate;
    if model = 'MODEL2' and variable = 'iv' then sa = StdErr;
    if model = 'MODEL3' and variable = 'mediator' then b = Estimate;
    if model = 'MODEL3' and variable = 'mediator' then sb = StdErr;
    if model = 'MODEL1' and variable = 'iv' then c = Estimate;
    if model = 'MODEL1' and variable = 'iv' then sc = StdErr;
    format model $model.;
run;

proc summary data=sobel nway;
    var a sa b sb c sc;
    output out=sobell max= ;
run;

/*****
/* Here we compute the sobel test, calculate the percentage of the total
/* effect that is mediated, calculate the ratio of the indirect to the direct
/* effect and create the final report
/* Further information about these tests, may be found in MacKinnon & Dwyer
/* (1993) Estimating mediated effects in prevention studies
*****/

data sobel2;
    set sobell;
    testest = (a*b)/sqrt(((b*b)*(sa*sa))+((a*a)*(sb*sb)));
    abssobel = abs(testest);
    p_val = 2*(1-CDF('NORMAL',abssobel));
    toteff = (a*b)/((a*b)+(c-(a*b)));
    ratio = (a*b)/((c-(a*b)));
    Test=1;
run;

/*****
/* GoodmanI test
*****/

data sobel3;
    set sobell;
    testest = (a*b)/sqrt(((b*b)*(sa*sa))+((a*a)*(sb*sb))+((sa*sa)*(sb*sb)));
    absgood = abs(testest);
    p_val = 2*(1-CDF('NORMAL',absgood));
    Test=2;
run;

/*****
/* GoodmanII test
*****/

data sobel4;
    set sobell;
    testest = (a*b)/sqrt(((b*b)*(sa*sa))+((a*a)*(sb*sb))-((sa*sa)*(sb*sb)));
    absgood2 = abs(testest);
    p_val = 2*(1-CDF('NORMAL',absgood2));
    Test=3;
run;

proc format;
    value Test
        1 = 'Sobel'
        2 = 'GoodmanI'
        3 = 'GoodmanII';
run;

```

```

data sobel5;
    set sobel2 sobel3 sobel4;          /*Concatenating Sobel and Goodman results*/
    format test test.;
run;

/*Creating a basic report to show the results*/

title1 'Results from regression analysis';
ods listing close;
ods html file = 'a:\mediation\sas\output\output1.html'
    style = sasweb;
proc report data=sobel nowd headline headskip
    style(header)=[background=skyblue]
    style(column)=[background=beige];
    column model variable Estimate StdErr tvalue probt;
    define model /order;
    define variable /display;
    define Estimate /display width=8 'Parameter Estimate' ;
    define StdErr /display width=8 'StdErr' ;
    define tvalue /display width=8 'tvalue' ;
    define probt /display width=8 'probt' ;
break after model / ol;
run;
title1 'Establishing Mediation';
title2 'Paths a,b, and c';
proc report data=sobel2 nowd headline headskip
    style(report)=[preimage="a:\mediation\sas\output\path.gif"]
    style(header)=[background=skyblue]
    style(column)=[background=beige];
    column a sa b sb c sc;
    define a /display width=8 'a' ;
    define sa /display width=8 'sa' ;
    define b /display width=8 'b' ;
    define sb /display width=8 'sb' ;
    define c /display width=8 'c' ;
    define sc /display width=8 'sc' ;
run;
title1 'Mediation Results';
proc report data=sobel5 nowd headline headskip
    style(header)=[background=skyblue]
    style(column)=[background=beige];
    column test testest p_val ;
    define test /order;
    define testest /display width=8 'Test statistic' ;
    define p_val /display width=8 'p-value' ;
break after test / ol;
run;
title1 'Percent Mediated';
proc report data=sobel5 nowd headline headskip
    style(header)=[background=skyblue]
    style(column)=[background=beige];
    column toteff ratio ;
    define toteff / analysis max format=percent8.2
        width=8 'Percent of the total effect that is mediated';
    define ratio / analysis max width=8 'Ratio of the indirect to the direct effect';
run;
ods html close;
ods listing;

```